```
!-----------------------------------------------------------------------
      attributes(global) subroutine gpuppush2l(ppart,fxy,kpic,qbm,dt,ek,&
     &idimp,nppmx,nx,ny,mx,my,nxv,nyv,mx1,mxy1,ipbc)
! for 2d code, this subroutine updates particle co-ordinates and
! velocities using leap-frog scheme in time and first-order linear
! interpolation in space, with various boundary conditions.
! threaded version using guard cells
! data read in tiles
! particles stored segmented array
! 44 flops/particle, 12 loads, 4 stores
! input: all, output: ppart, ek
! equations used are:
! vx(t+dt/2) = vx(t-dt/2) + (q/m)*fx(x(t),y(t))*dt,
! vy(t+dt/2) = vy(t-dt/2) + (q/m)*fy(x(t),y(t))*dt,
! where q/m is charge/mass, and
! x(t+dt) = x(t) + vx(t+dt/2)*dt, y(t+dt) = y(t) + vy(t+dt/2)*dt
! fx(x(t),y(t)) and fy(x(t),y(t)) are approximated by interpolation from
! the nearest grid points:
! fx(x,y) = (1-dy)*((1-dx)*fx(n,m)+dx*fx(n+1,m)) + dy*((1-dx)*fx(n,m+1)
!    + dx*fx(n+1,m+1))
! fy(x,y) = (1-dy)*((1-dx)*fy(n,m)+dx*fy(n+1,m)) + dy*((1-dx)*fy(n,m+1)
!    + dx*fy(n+1,m+1))
! where n,m = leftmost grid points and dx = x-n, dy = y-m
! ppart(n,1,m) = position x of particle n in tile m
! ppart(n,2,m) = position y of particle n in tile m
! ppart(n,3,m) = velocity vx of particle n in tile m
! ppart(n,4,m) = velocity vy of particle n in tile m
! fxy(1,j,k) = x component of force/charge at grid (j,k)
! fxy(2,j,k) = y component of force/charge at grid (j,k)
! that is, convolution of electric field over particle shape
! kpic = number of particles per tile
! qbm = particle charge/mass
! dt = time interval between successive calculations
! kinetic energy/mass at time t is also calculated, using
! ek = .125*sum((vx(t+dt/2)+vx(t-dt/2))**2+(vy(t+dt/2)+vy(t-dt/2))**2)
! idimp = size of phase space = 4
! nppmx = maximum number of particles in tile
! nx/ny = system length in x/y direction
! mx/my = number of grids in sorting cell in x/y
! nxv = first dimension of field arrays, must be >= nx+1
! nyv = second dimension of field arrays, must be >= ny+1
! mx1 = (system length in x direction - 1)/mx + 1
! mxy1 = mx1*my1, where my1 = (system length in y direction - 1)/my + 1
! ipbc = particle boundary condition = (0,1,2,3) =
! (none,2d periodic,2d reflecting,mixed reflecting/periodic)
      implicit none
      integer, value :: idimp, nppmx, nx, ny, mx, my, nxv, nyv
      integer, value :: mx1, mxy1, ipbc
      real, value :: qbm, dt
      real, dimension(nppmx,idimp,mxy1) :: ppart
      real, dimension(2,nxv,nyv) :: fxy
      integer, dimension(mxy1) :: kpic
      real, dimension(mxy1) :: ek
! local data
```

```
      integer :: noff, moff, npp, mxv
      integer :: i, j, k, ii, nn, mm
      real :: qtm, edgelx, edgely, edgerx, edgery
      real :: dxp, dyp, amx, amy, x, y, dx, dy, vx, vy
! The sizes of the shared memory arrays are as follows:
! float sfxy(2*(mx+1)*(my+1)), sek(blockDim%x)
! to conserve memory, sek overlaps with sfxy
! and the name sfxy is used instead of sek
      real, shared, dimension(*) :: sfxy
      double precision :: sum1
      qtm = qbm*dt
      sum1 = 0.0d0
! set boundary values
      edgelx = 0.0
      edgely = 0.0
      edgerx = real(nx)
      edgery = real(ny)
      if (ipbc==2) then
         edgelx = 1.0
         edgely = 1.0
         edgerx = real(nx-1)
         edgery = real(ny-1)
      else if (ipbc==3) then
         edgelx = 1.0
         edgerx = real(nx-1)
      endif
      mxv = mx + 1
! k = tile number
      k = blockIdx%x + gridDim%x*(blockIdx%y - 1)
! loop over tiles
      if (k <= mxy1) then
         noff = (k - 1)/mx1
         moff = my*noff
         noff = mx*(k - mx1*noff - 1)
         npp = kpic(k)
! load local fields from global array
         nn = min(mx,nx-noff) + 1
         mm = min(my,ny-moff) + 1
         ii = threadIdx%x
         do while (ii <= mxv*(my+1))
            j = (ii - 1)/mxv
            i = ii - mxv*j
            j = j + 1
            if ((i <= nn) .and. (j <= mm)) then
               sfxy(2*ii-1) = fxy(1,i+noff,j+moff)
               sfxy(2*ii) = fxy(2,i+noff,j+moff)
            endif
            ii = ii + blockDim%x
         enddo
! synchronize threads
         call syncthreads()
! loop over particles in tile
         j = threadIdx%x
         do while (j <= npp)
```

```
! find interpolation weights
            x = ppart(j,1,k)
            nn = x
            y = ppart(j,2,k)
            mm = y
            dxp = x - real(nn)
            dyp = y - real(mm)
            nn = 2*(nn - noff) + 2*mxv*(mm - moff) + 1
            amx = 1.0 - dxp
            amy = 1.0 - dyp
! find acceleration
            dx = amx*sfxy(nn)
            dy = amx*sfxy(nn+1)
            dx = amy*(dxp*sfxy(nn+2) + dx)
            dy = amy*(dxp*sfxy(nn+3) + dy)
            nn = nn + 2*mxv
            vx = amx*sfxy(nn)
            vy = amx*sfxy(nn+1)
            dx = dx + dyp*(dxp*sfxy(nn+2) + vx)
            dy = dy + dyp*(dxp*sfxy(nn+3) + vy)
! new velocity
            vx = ppart(j,3,k)
            vy = ppart(j,4,k)
            dx = vx + qtm*dx
            dy = vy + qtm*dy
! average kinetic energy
            vx = vx + dx
            vy = vy + dy
            sum1 = sum1 + dble(vx*vx + vy*vy)
            ppart(j,3,k) = dx
            ppart(j,4,k) = dy
! new position
            dx = x + dx*dt
            dy = y + dy*dt
! reflecting boundary conditions
            if (ipbc==2) then
               if ((dx < edgelx).or.(dx >= edgerx)) then
                  dx = ppart(j,1,k)
                  ppart(j,3,k) = -ppart(j,3,k)
               endif
               if ((dy < edgely).or.(dy >= edgery)) then
                  dy = ppart(j,2,k)
                  ppart(j,4,k) = -ppart(j,4,k)
               endif
! mixed reflecting/periodic boundary conditions
            else if (ipbc==3) then
               if ((dx < edgelx).or.(dx >= edgerx)) then
                  dx = ppart(j,1,k)
                  ppart(j,3,k) = -ppart(j,3,k)
               endif
            endif
! set new position
            ppart(j,1,k) = dx
            ppart(j,2,k) = dy
```

```
            j = j + blockDim%x
         enddo
! synchronize threads
         call syncthreads()
! add kinetic energies in tile
         sfxy(threadIdx%x) = real(sum1)
! synchronize threads
         call syncthreads()
         call lsum2(sfxy,blockDim%x)
! normalize kinetic energy of tile
         if (threadIdx%x==1) ek(k) = 0.125*sfxy(1)
      endif
      end subroutine
!
```

```
!-----------------------------------------------------------------------
      attributes(global) subroutine gpu2ppost2l(ppart,q,kpic,qm,nppmx,  &
     &idimp,mx,my,nxv,nyv,mx1,mxy1)
! for 2d code, this subroutine calculates particle charge density
! using first-order linear interpolation, periodic boundaries
! threaded version using guard cells
! data deposited in tiles
! particles stored segmented array
! 17 flops/particle, 6 loads, 4 stores
! input: all, output: q
! charge density is approximated by values at the nearest grid points
! q(n,m)=qm*(1.-dx)*(1.-dy)
! q(n+1,m)=qm*dx*(1.-dy)
! q(n,m+1)=qm*(1.-dx)*dy
! q(n+1,m+1)=qm*dx*dy
! where n,m = leftmost grid points and dx = x-n, dy = y-m
! ppart(n,1,m) = position x of particle n in tile m
! ppart(n,2,m) = position y of particle n in tile m
! q(j,k) = charge density at grid point j,k
! kpic = number of particles per tile
! qm = charge on particle, in units of e
! nppmx = maximum number of particles in tile
! idimp = size of phase space = 4
! mx/my = number of grids in sorting cell in x/y
! nxv = first dimension of charge array, must be >= nx+1
! nyv = second dimension of charge array, must be >= ny+1
! mx1 = (system length in x direction - 1)/mx + 1
! mxy1 = mx1*my1, where my1 = (system length in y direction - 1)/my + 1
      implicit none
      integer, value :: nppmx, idimp, mx, my, nxv, nyv, mx1, mxy1
      real, value :: qm
      real, dimension(nppmx,idimp,mxy1) :: ppart
      real, dimension(nxv,nyv) :: q
      integer, dimension(mxy1) :: kpic
! local data
      integer :: noff, moff, npp, mxv
      integer :: i, j, k, ii, nn, mm, np, mp
      real :: dxp, dyp, amx, amy, old
! The size of the shared memory array is as follows:
! real sq((mx+1)*(my+1))
      real, shared, dimension((mx+1)*(my+1)) :: sq
      mxv = mx + 1
! k = tile number
      k = blockIdx%x + gridDim%x*(blockIdx%y - 1)
! loop over tiles
      if (k <= mxy1) then
         noff = (k - 1)/mx1
         moff = my*noff
         noff = mx*(k - mx1*noff - 1)
         npp = kpic(k)
! zero out local accumulator
         i = threadIdx%x
         do while (i <= mxv*(my+1))
            sq(i) = 0.0
```

```
            i = i + blockDim%x
         enddo
! synchronize threads
         call syncthreads()
! loop over particles in tile
         j = threadIdx%x
         do while (j <= npp)
! find interpolation weights
            dxp = ppart(j,1,k)
            nn = dxp
            dyp = ppart(j,2,k)
            mm = dyp
            dxp = qm*(dxp - real(nn))
            dyp = dyp - real(mm)
            nn = nn - noff + 1
            mm = mxv*(mm - moff)
            amx = qm - dxp
            mp = mm + mxv
            amy = 1.0 - dyp
            np = nn + 1
! deposit charge within tile to local accumulator
! original deposit charge, has data hazard on GPU
!           sq(np+mp) = sq(np+mp) + dxp*dyp
!           sq(nn+mp) = sq(nn+mp) + amx*dyp
!           sq(np+mm) = sq(np+mm) + dxp*amy
!           sq(nn+mm) = sq(nn+mm) + amx*amy
! for devices with compute capability 2.x
            old = atomicAdd(sq(np+mp),dxp*dyp)
            old = atomicAdd(sq(nn+mp),amx*dyp)
            old = atomicAdd(sq(np+mm),dxp*amy)
            old = atomicAdd(sq(nn+mm),amx*amy)
            j = j + blockDim%x
         enddo
! synchronize threads
         call syncthreads()
! deposit charge to global array
         nn = min(mxv,nxv-noff)
         mm = min(my+1,nyv-moff)
         ii = threadIdx%x
         do while (ii <= mxv*(my+1))
            j = (ii - 1)/mxv
            i = ii - mxv*j
            j = j + 1
            if ((i <= nn) .and. (j <= mm)) then
! original deposit charge, has data hazard on GPU
!              q(i+noff,j+moff) = q(i+noff,j+moff) + sq(ii)
! for devices with compute capability 2.x
               old = atomicAdd(q(i+noff,j+moff),sq(ii))
            endif
            ii = ii + blockDim%x
         enddo
      endif
      end subroutine
```